Gamma calculus for singular knots, does it match Fiedler's old paper generalizing the Kauffman-state approach to long singular knots?

In[∘]:=

```
Γ /: Γ[ω1_, A1_] Γ[ω2_, A2_] := Γ[ω1 ω2, A1 + A2]
Γ /: Γ[ω1_, A1_] ≡ Γ[ω2_, A2_] := Simplify[ω1 - ω2 == 0 && A1 - A2 == 0]
ΓCollect[Γ[ω_, A_]] :=
 Γ[Simplify[ω] // Expand, Collect[A, r_, Collect[#, c_, Factor] &]]
ΓFormat[Γ[ω_, A_]] := Module[{S, M},
  S = Union@Cases[Γ[ω, A], (r | c)_a_ :> a, Infinity];
  M = Outer[Factor[∂_{c_#1, r_#2} A] &, S, S];
  M = Prepend[M, r_# & /@ S] // Transpose;
  M = Prepend[M, Prepend[c_# & /@ S, ω]];
  M // MatrixForm
 ]
X_{a_,b_} := Γ[1, r_a c_a + (1 - t) r_a c_b + t r_b c_b]
X̄_{a_,b_} := X_{a,b} /. t → t^-1
m_{a_,b_→k_}[Γ[ω_, A_]] :=
 Γ[(μ = 1 - ∂_{r_a} ∂_{c_b} A) ω, A + (∂_{r_a} A) (∂_{c_b} A) / μ] /. {(c_b | r_a) → 0, c_a → c_k, r_b → r_k} // ΓCollect

S_{a_,b_} := Γ[1, e c_a r_a + (1 - e t) c_b r_a + (1 - e) c_a r_b + e t c_b r_b]
```

Finding the value of the singular crossing $SS_{a,b}$ in Gamma calculus by the method of undetermined coefficients:

In[∘]:=

```
SS_{a_,b_} := Γ[1, e r_a c_a + f r_a c_b + g r_b c_a + h r_b c_b]
{sol} = Solve[Thread[Join[Flatten@Table[Coefficient[(SS_{1,2} X_{3,4} // m_{1,4→5} // m_{2,3→6})[[2]] -
        (SS_{1,2} X_{3,4} // m_{4,1→6} // m_{3,2→5})[[2]], r_v c_w], {v, {5, 6}}, {w, {5, 6}}],
     Flatten@Table[Coefficient[(SS_{1,2} X̄_{3,4} X_{5,6} // m_{1,6→1} // m_{4,2→2} // m_{5,3→3})[[2]] -
        (SS_{1,2} X_{3,4} X̄_{5,6} // m_{4,1→1} // m_{2,6→2} // m_{5,3→3})[[2]], r_v c_w],
       {v, {1, 2, 3}}, {w, {1, 2, 3}}],
     Flatten@Table[Coefficient[(SS_{1,2} X_{4,3} X̄_{6,5} // m_{1,6→1} // m_{4,2→2} // m_{5,3→3})[[2]] -
        (SS_{1,2} X̄_{4,3} X_{6,5} // m_{4,1→1} // m_{2,6→2} // m_{5,3→3})[[2]], r_v c_w], {v, {1, 2, 3}}, {w, {1, 2, 3}}]
    ] == 0], {e, f, g, h}]
SS_{a,b} /. sol // ΓFormat
```

Solve: Equations may not give solutions for all "solve" variables.

Out[∘]= $\{\{f \to 1 - e\,t,\ g \to 1 - e,\ h \to e\,t\}\}$

Out[∘]//MatrixForm=

$$\begin{pmatrix} 1 & c_a & c_b \\ r_a & e & 1 - e\,t \\ r_b & 1 - e & e\,t \end{pmatrix}$$

```
(* We call the general solution S_{a,b}.
   Notice e=1 is the positive crossing,
the negative crossing does not appear but its transpose does when e = t^-1*)
```

In[ ]:= $SS_{a,b}$ /. sol
$\{(S_{a,b}$ /. $e \to 1) \equiv X_{a,b}$, $(S_{a,b}$ /. $e \to t^{-1}) \equiv \overline{X}_{b,a}\}$

Out[ ]= $\Gamma[1, e\, c_a r_a + (1 - e\, t)\, c_b r_a + (1 - e)\, c_a r_b + e\, t\, c_b r_b]$

Out[ ]= {True, True}

In[ ]:= $S_{1,2}$ // $m_{1,2\to 1}$

Out[ ]= $\Gamma[h\, t, c_1 r_1]$

In[ ]:= $S_{1,2} S_{3,4}$ // $m_{1,2\to 1}$ // $m_{1,3\to 1}$ // $m_{1,4\to 1}$
$S_{4,1} S_{2,3}$ // $m_{1,2\to 1}$ // $m_{1,3\to 1}$ // $m_{1,4\to 1}$
$S_{1,3} S_{4,2}$ // $m_{1,2\to 1}$ // $m_{1,3\to 1}$ // $m_{1,4\to 1}$

Out[ ]= $\Gamma[h^2 t^2, c_1 r_1]$

Out[ ]= $\Gamma[h^2 t, c_1 r_1]$

Out[ ]= $\Gamma[1 - 2\, h\, t + h^2 t + h^2 t^2, c_1 r_1]$

In[ ]:= $S_{1,4} S_{5,2} X_{3,6}$ // $m_{1,2\to 1}$ // $m_{1,3\to 1}$ // $m_{1,4\to 1}$ // $m_{1,5\to 1}$ // $m_{1,6\to 1}$

Out[ ]= $\Gamma[t - 2\, h\, t^2 + h^2 t^2 + h^2 t^3, c_1 r_1]$

In[ ]:= $S_{1,2} S_{3,4}$ // $m_{1,4\to 1}$ // $m_{2,3\to 2}$

Out[ ]= $\Gamma[1, ((1 - 2\, h + h^2 + h^2 t)\, c_1 - h\, t\, (-2 + h + h\, t)\, c_2)\, r_1 +$
$(-h\, (-2 + h + h\, t)\, c_1 + (1 - 2\, h\, t + h^2 t + h^2 t^2)\, c_2)\, r_2]$

In[ ]:= $S_{1,4} X_{5,2} X_{3,6}$ // $m_{1,2\to 1}$ // $m_{1,3\to 1}$ // $m_{1,4\to 1}$ // $m_{1,5\to 1}$ // $m_{1,6\to 1}$

Out[ ]= $\Gamma[t - t^2 + h\, t^3, c_1 r_1]$

In[ ]:=
$Z = X_{8,1} S_{2,5} \overline{X}_{10,3} \overline{X}_{4,9} X_{6,11} X_{12,7}$; (*knot?*)
Do[$Z = Z$ // $m_{1,k\to 1}$, {k, 2, 12}]; Z

Out[ ]= $\Gamma[2 + h - \frac{1}{t} - 2\, t - h\, t + t^2 + 2\, h\, t^2 - h\, t^3, c_1 r_1]$